
Abstract. A shallow semantical embedding for public announcement
logic with relativized common knowledge is presented. This embedding
enables the first-time automation of this logic with off-the-shelf theorem
provers for classical higher-order logic. It is demonstrated (i) how meta-
theoretical studies can be automated this way, and (ii) how non-trivial
reasoning in the target logic (public announcement logic), required e.g. to
obtain a convincing encoding and automation of the wise men puzzle,
can be realized. Key to the presented semantical embedding—in con-
trast, e.g., to related work on the semantical embedding of normal modal
logics—is that evaluation domains are modeled explicitly and treated as
additional parameter in the encodings of the constituents of the embed-
ded target logic, while they were previously implicitly shared between
meta logic and target logic.

Keywords: Public announcement logic · Relativized common knowl-
edge · Semantical embedding · Higher-order logic · Proof automation

1 Introduction

Previous work has studied the application of a universal (meta-)logical reasoning
approach [5,6] for solving a prominent riddle in epistemic reasoning, known as
the wise men puzzle, on the computer [6]. The solution presented there puts a
particular emphasis on the adequate modeling of (ordinary) common knowledge
and it also illustrates the elegance and the practical relevance of the shallow
semantical embedding approach (in classical higher-order logic) [5], when be-
ing utilized within modern proof assistant systems such as Isabelle/HOL [20].
However, this work nevertheless falls short, since it did not convincingly address
the interaction dynamics between the involved agents. To do so, we extend and
adapt in this student paper the universal (meta-)logical reasoning approach for
public announcement logic and we demonstrate how it can be adapted to achieve
a convincing encoding and automation of the wise men puzzle in Isabelle/HOL
that also captures the interaction dynamics of the wise men puzzle scenario. In
more general terms, we present the first automation of public announcement
logic with relativized common knowledge, and we demonstrate that, and how,
this logic can be seen and elegantly handled as a fragment of classical higher-
order logic. Key to the presented extension of the shallow semantical embedding

approach is that the evaluation domains of the embedded target logic (public
announcement logic with relativized common knowledge) are no longer implic-
itly shared with the meta-logic (classical higher-order logic), but they are now
explicitly modeled as an additional parameter in the encoding of the embedded
logics constituents.

This paper is structured as follows: §2 briefly recaps classical higher-order
logic (Church’s type theory), and §3 sketches public announcement logic with
relativized common knowledge. The main contributions of this paper are then
presented in §4, where a shallow semantical embedding of public announcement
logic in classical higher-order logic is studied. In §5 the newly acquired embedding
is tested and applied to achieve an encoding and automation of the prominent
wise men puzzle. §6 discusses related work and §7 concludes the paper.

2 Classical Higher-Order Logic

We briefly recap classical higher-order logic (HOL), respectively Church’s simple
theory of types [11,7], which is a logic defined on top of the simply typed lambda
calculus. The presentation is partly adapted from Benzmüller [4]. For further
information on the syntax and semantics of HOL we refer to [8].

Syntax of HOL. We start out with defining the set T of simple types by the
following abstract grammar: α, β := o | i | (α → β). Type o denotes a bivalent set
of truth values, containing truth and falsehood, and i denotes a non-empty set of
individuals. Further base types are optional. → is the function type constructor,
such that (α → β) ∈ T whenever α, β ∈ T . We may generally omit parentheses.

The terms of HOL are defined by the following abstract grammar:

s, t := pα | Xα | (λxα.sβ)α→β | (sα→βtα)β

where α, β, o ∈ T . The pα ∈ Cα are typed constants and the Xα ∈ Vα are typed
variables (distinct from the pα). If sα→β and tα are HOL terms of types α → β
and α, respectively, then (sα→βtα)β , called application, is an HOL term of type
β. If Xα ∈ Vα is a typed variable symbol and sβ is an HOL term of type β, then
(λXαsβ)α→β , called abstraction, is an HOL term of type α → β. The type of each
term is given as a subscript. We call terms of type o formulas. As primitive logical
connectives we choose ¬o→o,∨o→o→o, =α→α→α and Π(α→o)→o. Other logical
connectives can be introduced as abbreviations; e.g. −→o→o→o= λyo.λyo.¬x∨y.

Semantics of HOL. A frame D for HOL is a collection {Dα}α∈T of nonempty
sets Dα, such that Do = {T, F} (for true and false). Di is chosen freely and
Dα→β are collections of functions mapping Dα into Dβ .

A model for HOL is a tuple M = 〈D, I〉, where D is a frame, and I is a
family of typed interpretation functions mapping constant symbols pα ∈ Cα to
appropriate elements of Dα, called the denotation of pα. The logical connectives

¬,∨, Π and = are always given their expected standard denotations:

I(¬o→o) = not ∈ Do→o s.t. not(T) = F and not(F) = T
I(∨o→o→o) = or ∈ Do→o→o s.t. or(a,b) = T iff (a = T or b = T)
I(=α→α→o) = id ∈ Dα→α→o s.t. for all a,b ∈ Dα, id(a,b) = T

iff a is identical to b.
I(Π(α→o)→o) = all ∈ D(α→o)→o s.t. for all s ∈ Dα→o, all(s) = T

iff s(a) = T for all a ∈ Dα

A variable assignment g maps variables Xα to elements in Dα. g[d/W ] denotes
the assignment that is identical to g, except for variableW , which is now mapped
to d.

The denotation JsαKM,g of an HOL term sα on a model M = 〈D, I〉 under
assignment g is an element d ∈ Dα defined in the following way:

JpαKM,g = I(pα)
JXαKM,g = g(Xα)
J(sα→βtα)βKM,g = Jsα→βKM,g(JtαKM,g)
J(λXαsβ)α→βKM,g = the function f from Dα to Dβ

s.t. f(d) = JsβKM,g[d/Xα] for all d ∈ Dα

In a standard model a domain Dα→β is defined as the set of all total functions
from Dα to Dβ: Dα→β = {f | f : Dα → Dβ}. In a Henkin model (or general
model) [15] function spaces are not necessarily required to be the full set of
functions: Dα→β ⊆ {f | f : Dα → Dβ}. However, we require that the valuation
function remains total, so that every term denotes (Denotatpflicht).

A HOL formula so is valid in an Henkin model M under assignment g if
and only if JsoK

M,g= T ; also denoted by M, g |=HOL so. An HOL formula so is
called valid in M, denoted by M |=HOL so, iff M, g |=HOL so for all assignments
g. Moreover, a formula so is called valid, denoted by |=HOL so, if and only if so is
valid in all Henkin models M.

Due to Gödel [14] a sound and complete mechanization of HOL with stan-
dard semantics cannot be achieved. For HOL with Henkin semantics sound and
complete calculi exist; cf. e.g. [9,8] and the references therein.

Each standard model is obviously also a Henkin model. Consequently, when
a HOL formula is Henkin-valid, it is also valid in all standard models.

3 Public Announcement Logic

The most important concepts and definitions of a public announcement logic
(PAL) with relativized common knowledge are depicted. For more details we
refer to [17,22].

Before exploring these definitions some general descriptions of the modeling
approach might be worthwhile. We use a graph-theoretical structure, called epis-
temic models, to represent knowledge. Epistemic models describe situations in

terms of possible worlds. A world represents one possibility about how the cur-
rent situation can be. Each agent is assumed to entertain a number of these pos-
sibilities. Knowledge is described using an accessibility relation between worlds,
rather than directly representing the agent’s information.

Let A be a set of agents and P a set of atomic propositions. Atomic propo-
sitions are intended to describe ground facts. We use a set W to denote possible
worlds and a valuation function V : P → ℘(W ) that assigns a set of worlds to
each atomic proposition. Vice versa, we may identify each world with the set of
propositions that are validated in them.

Definition 1 (Epistemic Model). Let A be a (finite) set of agents and P a
(finite or countable) set of atomic propositions. An Epistemic Model is a triple
M = 〈W, {Ri}i∈A, V 〉 where W 6= ∅, Ri ⊆ W ×W is an accessibility relation
(for each i ∈ A), and V : P → ℘(W ) is a valuation function (℘(W ) is the
powerset of W ).

Information of agent i at world w can now be defined as: Ri(w) = {v ∈
W | wRiv}. Having a separate (accessibility) relation for each agent enables
them to have their own viewpoints.

Next, we introduce the syntax of our base epistemic logic as the set of sen-
tences generated by the following grammar (where p ∈ P and i ∈ A):

ϕ, ψ := p | ¬ϕ | ϕ ∨ ψ | Kiϕ

We also introduce the abbreviations ϕ∧ψ := ¬(¬ϕ∨¬ψ) and ϕ→ ψ := ¬ϕ∨ψ.

Definition 2 (Truth at world w). Given an epistemic model M =
〈W, {Ri}i∈A, V 〉. For each w ∈ W,ϕ is true at world w, denoted M, w |= ϕ,
is defined inductively as follows:

M, w |= p iff w ∈ V (p)
M, w |= ¬ϕ iff M, w 6|= ϕ
M, w |= ϕ ∨ ψ iff M, w |= ϕ or M, w |= ψ
M, w |= Kiϕ iff for all v ∈W , if wRiv then M, v |= ϕ

The formula Kiϕ expresses that ”Agent i knows ϕ”. This describes knowledge
as an all-or-nothing definition. If we postulate that agent i knows ϕ, we say that
ϕ is true throughout all worlds in agents i’s range of considerations.

Satisfiabilty of a formula ϕ for a model M = 〈W, {Ri}i∈A, V 〉 and a world
w ∈ W is expressed by writing that M, w |= ϕ. We define VM(ϕ) = {w ∈
W | M, w |= ϕ}. Formula ϕ is valid if and only if for all M and for all worlds
w we have M, w |= ϕ.

Our modal logic above (corresponding to the normal modal logic K) is not
yet sufficiently suited to encode epistemic reasoning. Therefore, additional con-
ditions (reflexivity, transitivity and euclideaness) are imposed on the accessibil-
ity relations. This can e.g. be achieved by postulating the following principles,
resp. axiom schemata (in a Hilbert-style proof system).

Assumption Formula Property
T Truth Kiϕ→ ϕ Reflexive
4 Positive Introspection Kiϕ→ KiKiϕ Transitive
5 Negative Introspection ¬Ki → Ki¬Kiϕ Euclidean

We add public announcements [18] to our logic. The objective is to formulate
an operation that transforms the epistemic model such that all agents find out
that ϕ is true. This is achieved by taking the model M and discarding all worlds
in which ϕ is false. Afterwards all agents will only consider worlds in which ϕ is
true. Because of the publicity of the announcement all agents are aware of the
fact that all other agents know that ϕ holds true afterwards.

Definition 3 (Public Announcement). Suppose that M = 〈W, {Ri}i∈A, V 〉
is an epistemic model and ϕ is a formula (in the language of our base logic). After
all the agents find out that ϕ is true (i.e., ϕ is publicly announced), the resulting

model is M!ϕ = 〈W !ϕ, {R!ϕ
i }i∈A, V

!ϕ〉 where W !ϕ = {w ∈W | M, w |= ϕ},

R!ϕ
i = Ri ∩ (W !ϕ ×W !ϕ) for all i ∈ A, and V !ϕ(p) = V (p) ∩W !ϕ for all p ∈ P.
To say that ”ψ is true after the announcement of ϕ” is represented as [!ϕ]ψ.

Truth for this new operator is defined as:

M, w |= [!ϕ]ψ iff M, w 6|= ϕ or M!ϕ, w |= ψ

We conclude this section with the introduction of notions for group knowledge.
Mutual knowledge, often stated as everyone knows, describes knowledge that

each member of the group holds. Usually, it is defined for a group of agentsG ⊆ A
as EGϕ :=

∧
i∈GKiϕ. Equivalently, a new relation can be introduced to express

mutual knowledge with the knowledge operator.

Definition 4 (Mutual Knowledge). Let G ⊆ A be a group of agents. Let
RG =

⋃
i∈GRi. The truth clause for mutual knowledge is:

M, w |= EGψ iff for all v ∈ W , if wRGv then M, v |= ψ

Still, there is a distinction to make between everyone knows ϕ and it is common
knowledge that ϕ. A statement p is common knowledge when all agents know p,
know that they all know p, know that they all know that they all know p, and so
ad infinitum. Relativized common knowledge was introduced by van Benthem,
van Eijck and Kooi [3] as a variant of common knowledge for dynamic epistemic
logics. As the name suggests knowledge update is then treated as a relativization.

Definition 5 (Relativized Common Knowledge). Let G ⊆ A be a group of
agents. Let RG =

⋃
i∈GRi. The truth clause for relativized common knowledge

is:
M, w |= CG(ϕ|ψ) iff for all v ∈W , if w(Rϕ

G)
+v then M, v |= ψ

where Rϕ
G = RG ∩ (W×VM(ϕ)), and (Rϕ

G)
+ denotes the transitive closure of

Rϕ
G.

Intuitively, CG(ϕ|ψ) expresses, that after ϕ is announced, ψ becomes common
knowledge in the group. This means, that every path from w, that is accessible
using the agent’s relations through worlds in which ϕ is true, must end in a
world in which ψ is true. Ordinary common knowledge of ϕ can be abbreviated
as CG(⊤|ϕ), where ⊤ denotes an arbitrary tautology.

In the remainder we use PAL to refer to the depicted logic consisting of modal
logic K, extended by the principles T45, public announcement and relativized
common knowledge.

4 Modeling PAL as a Fragment of HOL

A shallow semantical embedding (SSE) of a target logic into HOL provides
a translation between the two logics in such a way that the former logic is
identified and characterized as a proper fragment of the latter. Once such an
SSE is obtained, all that is needed is to prove (or refute) conjectures in the
target logic is to provide the SSE, encoded in an input file, to the HOL prover
in addition to the encoded conjecture. We can then use the HOL prover as-is,
without making any changes to its source code, and use it to solve problems in
our target logic.

4.1 Shallow Semantical Embedding

To define an SSE for target logic PAL we lift the type of propositions in order
to explicitly encode their dependency on possible worlds; this is analogous to
prior work [5,6]. In order to capture the model-changing behavior of PAL we
additionally introduce world domains (sets of worlds) as parameters/arguments
in the encoding. The rationale thereby is to be able to suitably constrain, and
recursively pass-on, these domains after each model changing action.

PAL formulas are thus identified in our semantical embedding with certain
HOL terms (predicates) of type (i→ o) → i→ o. They can be applied to terms
of type i→ o, which are assumed to denote evaluation domains, and subse-
quently to terms of type i, which are assumed to denote possible worlds. That
is, the HOL type i is identified with a (non-empty) set of worlds, and the type
i→ o, abbreviated by σ, is identified with a set of sets of worlds, i.e., a set of
evaluation domains. Type (i→ o) → i→ o is abbreviated as τ , and type α is an
abbreviation for i→ i→ o, the type of accessibility relations between worlds.

For each propositional symbol pi of PAL, the corresponding HOL signature
is assumed to contain a corresponding constant symbol piσ, which is (rigidly)
denoting the set of all those worlds in which pi holds. We call the piσ σ-type-
lifted propositions. Moreover, for k = 1, . . . , |A| the HOL signature is assumed to

contain the constant symbols r1α, . . . , r
|A|
α . Without loss of generality, we assume

that besides those constants symbols and the primitive logical connectives of
HOL, no other constant symbols are given in the signature of HOL.

As a simplifying assumption in this ongoing work (which has a particular
focus on an automation of the Wise Men Puzzle in PAL) we continue with
choosing |A| = 3. (A generalization for arbitrary A is straightforward).

The mapping ⌊·⌋ translates a formula ϕ of PAL into a term ⌊ϕ⌋ of HOL of
type τ . The mapping is defined recursively:

⌊pj⌋ = (A(pjσ))τ

⌊¬ϕ⌋ = ¬τ→τ ⌊ϕ⌋

⌊ϕ ∨ ψ⌋ = ∨τ→τ→τ⌊ϕ⌋⌊ψ⌋

⌊K rk ϕ⌋ = Kα→τ→τ rkα ⌊ϕ⌋

⌊[!ϕ]ψ⌋ = [! · ] ·τ→τ→τ ⌊ϕ⌋⌊ψ⌋

⌊C(ϕ|ψ)⌋ = C(·|·)τ→τ→τ⌊ϕ⌋⌊ψ⌋

Operator A(·), which evaluates atomic formulas, is defined as follows:

A·σ→τ = λAσλDσλXi(D X ∧ A X)

As a first argument it accepts a σ-type-lifted proposition Aσ, which are rigidly
interpreted. As a second argument it accepts an evaluation domain Dσ, that
is, an arbitrary subset of the domain associated with type σ. And as a third
argument it accepts a current world. It then checks whether (i) the current
world is a member of evaluation domain Dσ and (ii) whether the σ-type-lifted
proposition Aσ holds in the current world.

The other logical connectives of PAL, except for [! · ]·τ→τ→τ , are now defined
in a way so that they simply pass-on the evaluation domains as parameters to the
atomic-level. Only [! · ]·τ→τ→τ is modifying, in fact, constraining, the evaluation
domain it passes on, and it does this in the expected way (cf. Def. 3):

¬τ→τ = λAτλDσλXi¬(A D X)

∨τ→τ→τ = λAτλBτλDσλXi(A D X ∨B D X)

Kα→τ→τ = λRαλAτλDσλXi∀Yi((D Y ∧ R X Y ) −→ A D Y )

[! · ]·τ→τ→τ = λAτλBτλDσλXi(¬(A D X) ∨ (B (λYi D Y ∧ A D Y ) X))

To model C(·|·)τ→τ→τ we reuse the following operations on relations; cf. [5,6].

transitiveα→o = λRα∀Xi∀Yi∀Zi(¬(R X Y ∧ R Y Z) ∨ R X Z)

intersectionα→α→α = λRαλQαλXiλYi(R X Y ∧ Q X Y )

unionα→α→α = λRαλQαλXiλYi(R X Y ∨ Q X Y )

subα→α→o = λRαλQα∀Xi∀Yi(¬R X Y ∨ Q X Y )

tcα→α = λRαλXiλYi∀Qα

(¬transitive Q ∨ (¬sub R Q ∨ Q X Y ))

Additionally, EVR is defined as the union of three agents r1, r2 and r3 of type
α. EVR can then be used as a relation, e.g., for the knowledge operator to de-
scribe mutual knowledge of the three agents. But most importantly, we need this
relation in order to encode relativized common knowledge.

EVRα = union(union r1 r2) r3

We want to remark that a general higher-order definition for the union of a
set of relations could alternatively be introduced first and then be applied to
our concrete set of relations R consisting of r1, r2 and r3. Nothing prevents
us from generalizing the notion of mutual knowledge this way to an arbitrary
group of agents R, and to consider R as a further parameter in e.g. the definition
of C(·|·)τ→τ→τ . However, in our first experiments as presented in this student
paper, which are primarily intended to study the practical feasibility of the
embedding approach for PAL, we have still avoided this final generalization
step. The operator C(·|·)τ→τ→τ thus abbreviates the following HOL term:

C(·|·)τ→τ→τ = λAτλBτλDσλXi∀Yi

(tc(intersection EVR (λUiλVi(D V ∧ A D V ))) X Y

−→ B D Y )

Analyzing the truth of a PAL formula ϕ, represented by the HOL term ⌊ϕ⌋, in
a particular domain d, represented by the term Dσ, and a world s, represented
by the term Si, corresponds to evaluating the application (⌊ϕ⌋ Dσ Si). ϕ is thus
generally valid if and only if for all Dσ and all Si we have D S → ⌊ϕ⌋D S.

The validity function, therefore, is defined as follows:

vldτ→o = λAτ∀Dσ∀Si(D S −→ A D S).

The necessity to quantify over all possible domains in this definition will be
further illustrated below.

4.2 Encoding into Isabelle/HOL

What follows is a description of the concrete encoding of the presented SSE of
PAL in HOL within the higher-order proof assistant Isabelle/HOL.1

All necessary types can be modeled in a straightforward way. We declare i

to denote possible worlds and then introduce type aliases for σ, τ and α. Type
bool represents (the bivalent set of) truth values.

typedecl i

type_synonym σ = "i⇒bool"

type_synonym τ = "σ ⇒i⇒bool"

type_synonym α = "i⇒i⇒bool"

The agents are declared mutually distinct accessibility relations and the group
of agents is denoted by predicate A. In order to obtain S5 (KT45) properties, we
declare respective conditions on the accessibility relations in the group of agents
A. Various Isabelle/HOL encodings from [5,6] are reused here (without mention-
ing due to space restrictions), including the encoding of transitive closure.

1 The full sources of our encoding can be found at
http://logikey.org in subfolder Public-Announcement-Logic, resp. at
https://github.com/cbenzmueller/LogiKEy/tree/master/Public-Announcement-Logic.

http://logikey.org
https://github.com/cbenzmueller/LogiKEy/tree/master/Public-Announcement-Logic

consts a::"α” b::"α" c::"α"

abbreviation "A x ≡ x = a ∨ x = b ∨ x = c"

axiomatization where

alldifferent: "¬(a = b) ∧ ¬(a = c) ∧ ¬(b = c)" and

agents_S5: "∀i. A i −→ (reflexive i ∧ transitive i ∧ euclidean i)"

abbreviation EVR :: "α" ("EVR")

where "EVR ≡ union_rel (union_rel a b) c"

To distinguish between HOL connectives (e.g. ¬) and the lifted PAL connectives
(e.g.¬τ→τ ) we make use of bold face fonts, see for example the definition¬τ→τ ≡
λϕτ .λWσ .λwi.¬ϕ W w below. Each of the lifted unary and binary connectives of
PAL accepts arguments of type τ , i.e. lifted PAL formulas, and returns such a
lifted PAL formula.

A special case, as discussed before, is the new operator for atomic propositions
A(·). When evaluating σ-type lifted atomic propositions p we need to check if p
is true in the given world w, but we also need to check whether the given world
w is still part of our evaluation domain W that has been recursively passed-on.
Operator A(·) is thus of type ”σ ⇒ τ”.

abbreviation patom :: "σ ⇒ τ" ("A_")

where "Ap ≡ λW w. W w ∧ p w"

abbreviation ptop :: τ ("⊤")

where "⊤ ≡ λW w. True"

abbreviation pneg :: "τ⇒τ" ("¬")

where "¬ϕ ≡ λW w. ¬(ϕ W w)"

abbreviation pand :: "τ⇒τ⇒τ" ("∧")

where "ϕ∧ ψ ≡ λW w. (ϕ W w) ∧ (ψ W w)"

abbreviation por :: "τ⇒τ⇒τ" ("∨")

where "ϕ∨ ψ ≡ λW w. (ϕ W w) ∨ (ψ W w)"

abbreviation pimp :: "τ⇒τ⇒τ" ("→")

where "ϕ → ψ ≡ λW w. (ϕ W w) −→ (ψ W w)"

abbreviation pequ :: "τ⇒τ⇒τ" ("↔")

where "ϕ ↔ ψ ≡ λW w. (ϕ W w) ←→ (ψ W w)"

In the definition of the knowledge operator K, we have to make sure to add a
domain check in the implication.

abbreviation pknow :: "τ⇒τ⇒τ" ("K _ _") modified
where "K r ϕ ≡ λW w.∀v. (W v ∧ r w v) −→ (ϕ W v)"

Two additional abbreviations are introduced to improve readability. A more
concise way to state knowledge and an additional operator for mutual knowledge,
in which the EVR relation gets used.

abbreviation agtknows :: "τ⇒τ⇒τ" ("K _")

where "Kr ϕ ≡ K r ϕ"

abbreviation evrknows :: "τ⇒τ" ("EA _")

where "EA ϕ ≡ K EVR ϕ"

We finally see the change of the evaluation domain in action, when introduc-
ing the public announcement operator. We already inserted domain checks in
the definition of the operators K and A(·). Now, we need to constrain the do-
main after each public announcement. So far the evaluation domain, modeled
by W, got passed-on through all lifted operators without any change. In the pub-
lic announcement operator, however, we modify the evaluation domain W into
(λz. W z ∧ ϕ W z) (i.e., the set of all worlds z in W, such that ϕ holds for W

and z), which is then recursively passed-on. The public announcement operator
is thus defined as:

abbreviation ppal :: "τ⇒τ⇒τ" ("[!_]_")

where "[!ϕ]ψ ≡ λW w. ¬(ϕ W w) ∨ (ψ (λz. W z ∧ ϕ W z) w)"

The following embedding of relativized common knowledge is a straightforward
encoding of the semantic properties and definitions as proposed in Def. 5.

abbreviation prck :: "τ⇒τ⇒ τ" ("CL_|_M")

where "CLϕ|ψM" ≡ λW w. ∀v.

(tc (intersection_rel EVR (λu v. W v ∧ ϕ W w)) w v) −→ (ψ W v)"

As described earlier we can abbreviate ordinary common knowledge as CG(⊤|ϕ):

abbreviation pcmn :: "τ⇒τ" ("CA _") where "CA ϕ ≡ CL⊤|ϕM"

Finally an embedding for the notion of validity is needed. Generally, for a type-
lifted formula ϕ to be valid, the application of ϕ to w has to hold true for
all worlds w. In the context of PAL the evaluation domains also have to be
incorporated in the definition. Originally we were tempted to define PAL validity
in such that we start with a ”full evaluation domain”, a domain that evaluates
to True for all possible worlds and gets restricted, whenever necessary after an
announcement. Such a validity definition would look like this:

abbreviation tvalid::"τ⇒bool" ("⌊_⌋T") where "⌊_⌋T ≡ ∀w. ϕ (λx. True) w"

But this leads to undesired behavior, which we can easily see when using our
reasoning tools to study e.g. the validity of an often proposed schematic axiom
of PAL, Announcement Necessitation: from ψ, infer [!ϕ]ψ. If we check for a
counterexample in Isabelle/HOL, the model finder Nitpick reports the following:

lemma necessitation: assumes "⌊ψ⌋T" shows "⌊[!ϕ]ψ⌋T" nitpick oops

Nitpick found a counterexample for card i = 2:

Free variables:

ϕ = (λx. _)

(((λx. _)(i1 := True, i2 := True), i1) := False,

((λx. _)(i1 := True, i2 := True), i2) := True,

((λx. _)(i1 := True, i2 := False), i1) := False,

((λx. _)(i1 := True, i2 := False), i2) := False,

((λx. _)(i1 := False, i2 := True), i1) := False,

((λx. _)(i1 := False, i2 := True), i2) := False,

((λx. _)(i1 := False, i2 := False), i1) := False,

((λx. _)(i1 := False, i2 := False), i2) := False)

ψ = (λx. _)

(((λx. _)(i1 := True, i2 := True), i1) := True,

((λx. _)(i1 := True, i2 := True), i2) := True,

((λx. _)(i1 := True, i2 := False), i1) := False,

((λx. _)(i1 := True, i2 := False), i2) := False,

((λx. _)(i1 := False, i2 := True), i1) := False,

((λx. _)(i1 := False, i2 := True), i2) := False,

((λx. _)(i1 := False, i2 := False), i1) := False,

((λx. _)(i1 := False, i2 := False), i2) := False)

Skolem constant:

??.tvalid.w = i2

The valid function needs instead to be defined such that it checks validity not
only for all worlds, but for all domains and worlds. Otherwise, the observed but
undesired value flipping may occur.

abbreviation pvalid :: "τ⇒bool" ("⌊_⌋")

where "⌊_⌋ ≡ ∀W.∀w. W w −→ ϕ W w "

All here introduced definitions are hidden from the user, who can construct
formulas in PAL and prove these using the newly embedded operators.

5 Experiments

5.1 Proving Axioms and Rules of Inference of PAL in HOL

The presented SSE of PAL is able to prove the following axioms and rules of
inference as presented for PAL in [2, see also Appendix F]:

System K
– All substitutions instances of propositional tautologies
Axiom K Ki(ϕ→ ψ) → (Kiϕ→ Kiψ)
Modus ponens From ϕ and ϕ→ ψ infer ψ
Necessitation From ϕ infer Kiϕ

System S5
Axiom T Kiϕ→ ϕ
Axiom 4 Kiϕ→ KiKiϕ
Axiom 5 ¬Kiϕ→ Ki¬Kiϕ

Reduction Axioms
Atomic Permanence [!ϕ]p↔ (ϕ→ p)
Conjunction [!ϕ](ψ ∧ χ) ↔ ([!ϕ]ψ ∧ [!ϕ]χ)
Partial Functionality [!ϕ]¬ψ ↔ (ϕ→ ¬[!ϕ]ψ)
Action-Knowledge [!ϕ]Kiψ ↔ (ϕ→ Ki(ϕ→ Ki(ϕ→ [!ϕ]ψ)))
– [!ϕ]C(χ|ψ) ↔ (ϕ→ C(ϕ ∧ [!ϕ]χ|[!ϕ]ψ))



12 S. Reiche and C. Benzmüller

Axiom schemes for RCK
C-normality C(χ|(ϕ→ ψ)) → (C(χ|ϕ) → C(χ|ψ))
Mix axiom C(ψ|ϕ) ↔ E(ψ → (ϕ ∧ C(ψ|ϕ)))
Induction axiom (E(ψ → ϕ) ∧ C(ψ|ϕ→ E(ψ → ϕ))) → C(ψ|ϕ)

Rules of Inference
Announcement Nec. from ϕ, infer [!ψ]ϕ
RKC Necessitation from ϕ, infer C(ψ|ϕ)

Only for the mix- and induction axiom (schemata) for relativized common knowl-
edge is one direction, respectively, not automatically provable yet. Structural
induction is required and a proof still needs to be provided by hand.

(*System K*)

lemma tautologies: "⌊⊤⌋" by auto

lemma axiom_K: "A i =⇒ ⌊(Ki (ϕ→ ψ)) → ((Ki ϕ) → (Ki ψ))⌋" by auto

lemma modusponens: assumes 1: "⌊ϕ → ψ⌋" and 2: "⌊ϕ⌋" shows "⌊ψ⌋"

using 1 2 by auto

lemma necessitation: assumes 1: "⌊ϕ⌋" shows "A i =⇒ ⌊Ki ϕ⌋"

using 1 by auto

(*More axiom systems*)

lemma axiom_T: "A i =⇒ ⌊(Ki ϕ) → ϕ)⌋"

using group_S5 reflexive_def by auto

lemma axiom_4: "A i =⇒ ⌊(Ki ϕ) → (Ki (Ki ϕ))⌋"

by (meson group_S5 transitive_def)

lemma axiom_5: "A i =⇒ ⌊(¬Ki ϕ) → (Ki (¬Ki ϕ))⌋

by (meson euclidean_def group_S5)

(*Reduction Axioms*)

lemma atomic_permanence: "⌊([!ϕ]Ap) → (ϕ→A p)⌋ by auto

lemma conjunction: "⌊([!ϕ](ψ ∧ χ)) ↔ (([!ϕ]ψ)∧ ([!ϕ]χ))⌋ by auto

lemma partial_functionality: "⌊([!ϕ]¬ψ) ↔ (ϕ→ (¬[!ϕ]ψ))⌋ by auto

lemma action_knowledge: "A i =⇒ ⌊([!ϕ](Ki ψ)) ↔ (ϕ→ (Ki (ϕ→ (([!ϕ]ψ))))⌋

by auto

lemma "⌊([!ϕ]CLψ|χM) ↔ (ϕ→ CL[!ϕ]ψ|[!ϕ]χM)⌋

by (smt intersection_rel_def sub_rel_def tc_def transitive_def)

(*Axiom schemes for RCK*)

lemma C_normality: "⌊(CLχ|ϕ → ψM) → (CLχ|ϕ M → CLχ|ψM)⌋

unfolding Defs by blast

lemma mix_axiom1: "⌊CLχ|ϕ M → (EA(χ→ (ϕ∧ CLχ|ψM)))⌋

unfolding Defs by metis

lemma mix_axiom2: "⌊(EA (χ → (ϕ∧ CLχ|ψM))) → CLχ|ϕM⌋

unfolding Defs sledgehammer (*timeout*)

lemma induction_axiom1: "⌊(EA (χ→ ϕ))∧ CLχ|ϕ → (EA (χ→ ϕ))M) → CLχ|ϕM⌋

unfolding Defs sledgehammer (*timeout*)

lemma induction_axiom2: "⌊CLχ|ϕ M → (EA (χ→ ϕ))∧ CLχ|ϕ → (EA (χ→ ϕ))M)⌋

unfolding Defs by smt

(*Rules of Inference*)

lemma announcement_nec: assumes 1: "⌊ϕ⌋" shows "⌊[!ψ]ϕ⌋" using 1 by auto

lemma rkc_necessitation: assumes 1: "⌊ϕ⌋" shows "⌊CLχ|ϕM⌋"

using 1 by (metis intersection_rel_def sub_rel_def tc_def transitive_def)

5.2 Exploring Failures of Uniform Substitution

The following principles are examples of sentences that are valid for eternal
sentences p, but not schematically valid [16].

1. p→ ¬[!p](¬p)

lemma "⌊Ap → ¬[!Ap](¬Ap)⌋ by simp

lemma "⌊ϕ → ¬[!ϕ](¬ϕ)⌋ nitpick oops (*countermodel found*)

2. p→ ¬[!p](¬Kip)

lemma "⌊Ap → ¬[!Ap](¬Ka
Ap)⌋ by simp

lemma "⌊ϕ → ¬[!ϕ](¬Ka ϕ)⌋ nitpick oops (*countermodel found*)

3. p→ ¬[!p](p ∧ ¬Kip)

lemma "⌊Ap → ¬[!Ap](Ap ∧¬ Ka
Ap)⌋ by simp

lemma "⌊ϕ → ¬[!ϕ](ϕ∧ ¬Ka ϕ)⌋ nitpick oops (*countermodel found*)

4. (p ∧ ¬Kip) → ¬[!p ∧ ¬Kip](p ∧ ¬Kip)

lemma "⌊(Ap∧¬Ka
Ap) → ¬[!Ap ∧¬Ka

Ap](Ap∧¬Ka
Ap)⌋ by blast

lemma "⌊(ϕ∧¬Ka ϕ) → ¬[!ϕ∧¬Ka ϕ](ϕ∧¬Ka ϕ)⌋ nitpick oops (*ctm. fd.*)

5. Kip→ ¬[!p](¬Kip)

lemma "⌊(Ka
Ap) → ¬[!Ap](¬Ka

Ap)⌋ using group_S5 reflexive_def by auto

lemma "⌊(Kaϕ) → ¬[!ϕ](¬Ka ϕ)⌋ nitpick oops (*countermodel found*)

6. Kip→ ¬[!p](p ∧ ¬Kip)

lemma "⌊(Ka
Ap) → ¬[!Ap](Ap∧¬Ka

Ap)⌋ using group_S5 reflexive_def by auto

lemma "⌊(Ka ϕ) → ¬[!ϕ](ϕ∧¬Ka ϕ)⌋ nitpick oops (*countermodel found*)

5.3 Example Application: The Wise Men Puzzle

The Wise Men puzzle is a interesting riddle in epistemic reasoning. It is well
suited to demonstrate epistemic actions in a multi-agent scenario. Baldoni [1]
gave a formulation for this, which later got embedded into Isabelle/HOL by
Benzmüller [5,6]. In the following implementation these results will be used as a
stepping stone.

First the riddle is recited, and then we go into detail on how the uncertainties
of all three agents change. The reader is invited to try to solve the riddle on her
own before continuing with the analysis.

Once upon a time, a king wanted to find the wisest out of his three wisest men.
He arranged them in a circle and told them that he would put a white or a
black spot on their foreheads and that one of the three spots would certainly be
white. The three wise men could see and hear each other but, of course, they

could not see their faces reflected anywhere. The king, then, asked each of them
[sequentially] to find out the color of his own spot. After a while, the wisest
correctly answered that his spot was white.

The already existing encoding by Benzmüller puts a particular emphasis on the
adequate modeling of common knowledge. Here, this solution will be enhanced
by the public announcement operator. Consequently, common knowledge will
not be statically stated after each iteration, but a dynamic approach is used for
this.

Before we can evaluate the knowledge of the first wise man we need to formu-
late the initial circumstances and background knowledge. Let a, b and c be the
wise men. It is common knowledge, that each wise man can see the foreheads of
the other wise men. The only doubt a wise man has, is whether he has a white
spot on his own forehead or not. Additionally, it is common knowledge that at
least one of the three wise men has a white spot on his forehead. The rules of
the riddle are embedded as follows:2

consts ws :: "α⇒ σ"

axiomatization where

(* Common knowledge: at least one of a, b and c has a white spot *)

WM1: "⌊CA (Aws a ∨ Aws b ∨ Aws c)⌋"
(* Common knowledge: if x has not a white spot then y know this *)

WM2ab: "⌊CA (¬(Aws a) → Kb(¬(Aws a)))⌋"
WM2ac: "⌊CA (¬(Aws a) → Kc(¬(Aws a)))⌋"
WM2ba: "⌊CA (¬(Aws b) → Ka(¬(Aws b)))⌋"
WM2bc: "⌊CA (¬(Aws b) → Kc(¬(Aws b)))⌋"
WM2ca: "⌊CA (¬(Aws c) → Ka(¬(Aws c)))⌋"
WM2cb: "⌊CA (¬(Aws c) → Kb(¬(Aws c)))⌋"

Now the king asks a whether he knows if he has a white spot or not. Assume
that a publicly answers that he does not. This is a public announcement of the
form: ¬(Ka(Aws a)) ∨ Ka¬(Aws a)). Again, a wise man gets asked by the king
whether he knows if he has a white spot or not. Now its b’s turn and assume
that b also announces that he does not know whether he has a white spot on his
forehead.3

When asked, c is able to give the right answer, namely that he has a white
spot on his forehead. We can prove this automatically in Isabelle/HOL:

theorem whitespot_c :

"⌊[!¬((Ka(
Aws a))∨ (Ka(¬(Aws a)))]([!¬((Kb(

Aws b))∨ (Kb(¬(Aws b)))](Kc(
Aws c)))⌋"

using WM1 WM2ba WM2ca WM2cb group_S5

unfolding reflexive_def intersection_rel_def

union_rel_def sub_rel_def tc_def

by smt

2 One might also add axioms of the form ⌊CA (Aws x) → Ky(
Aws x))⌋” for x, y ∈ A.

This is not necessary as we will see in the proof found using Isabelle/HOL.
3 The case where neither a nor b can correctly infer the color of their forehead when
being asked by the king is the most challenging case; we only discuss this one here.

6 Comparison with Related Work

In related work [21], van Benthem, van Eijck and colleagues have studied a
“faithful representation of DEL [dynamic epistemic logic] models as so-called
knowledge structures that allow for symbolic model checking”. The authors show
that such an approach enables efficient and effective reasoning in epistemic sce-
narios with state-of-the-art Binary Decision Diagram (BDD) reasoning technol-
ogy, outperforming other existing methods [24,25] to automate DEL reasoning.
Further related work [23] demonstrates how dynamic epistemic terms can be
formalized in temporal epistemic terms to apply the model checkers MCK [12]
or MCMAS [19]. Our approach differs in various respects, incuding:

External vs. internal representation transformation: Instead of writing
external (e.g Haskell-)code to realize the required conversions from DEL into
Boolean representations, we work with logic-internal conversions into HOL,
provided in form of a set of equations stated in HOL itself (thereby heavily
exploiting the virtues of λ-abstraction and λ-conversion). Our encoding is
concise (only about 50 lines in Isabelle/HOL) and human readable.

Meta-logical reasoning: Since our conversion “code” is provided within the
(meta-)logic environment itself, the conversion becomes better controllable
and even amenable to formal verification. Moreover, as we have also demon-
strated in this paper, meta-logical studies about the embedded logics and
their embedding in HOL are well-supported in our approach.

Scalability beyond propositional reasoning: Real world applications often
require differentiation between entities/individuals, their properties and func-
tions defined on them, and quantification over entities, or even properties
and functions, supports generic statements that are not supported in propo-
sitional DEL. The shallow semantical embedding approach, in contrast, very
naturally scales for first-order and higher-order extensions of the embedded
logics; for more details on this we refer to [5,6] and the references therein.

Reuse of automated theorem proving and model finding technology:
Both approaches reuse state-of-the-art automated reasoning technology. In
our case this includes world-leading first-order and higher-order theorem
provers and model finders already integrated with Isabelle/HOL [10]. These
tools in turn internally collaborate with latest SMT and SAT solving tech-
nology. The burden to organize and orchestrate the technical communication
with and between these tools is taken away from us by reuse of respective
solutions as already provided in Isabelle/HOL (and recursively also within
the integrated theorem provers). Well established and robustly supported
language formats (e.g. TPTP syntax, http://www.tptp.org) are reused
in these nested transformations. These cascades of already supported logic
transformations are one reason why our embedding approach readily scales
for automating reasoning beyond just propositional DEL.

We are convinced, as evidenced by the above discussion, that our approach is
particularly well suited for the exploration and rapid prototyping of new logics

http://www.tptp.org

(and logic combinations) and their embeddings in HOL, and for the study of their
meta-logical properties, in particular, when it comes to first-order and higher-
order extensions of DEL. At the same time we share with the related work by van
Benthem, van Eijck a and colleagues a deep interest in practical (object-level)
applications, and therefore practical reasoning performance is obviously also
of high relevance. In this regard, however, we naturally assume a performance
loss in comparison to hand-crafted, specialist solutions. Previous studies in the
context of first-order modal logic theorem proving nevertheless have shown that
this is not always the case [13]. Future work therefore includes the conduction
of comparative performance studies in which the work presented in this paper
is compared with the existing alternative approaches.

7 Conclusion

A shallow semantical embedding of public announcement logic with relativized
common knowledge in classical higher-order logic has been presented, and our
implementation of this embedding in Isabelle/HOL delivers results as expected.
In particular, we have shown how model-changing behaviour can be adequately
and elegantly addressed in our embedding approach. With reference to uniform
substitution, we saw that our embedding enables the study of meta-logical prop-
erties of public announcement logic, and object-level reasoning has been demon-
strated by a first time automation of the wise men puzzle encoded in public
announcement logic with a relativized common knowledge operator.

Further work includes the provision of proofs for the faithfulness of the pre-
sented embedding; this should be analogous to prior work, see e.g. [4].

Acknowledgments We thank David Streit, David Fuenmayor and the anony-

14. K. Gödel. Über formal unentscheidbare Sätze der Principia Mathematica und
verwandter Systeme I. Monatshefte für Mathematik und Physik, 38(1):173–198,
1931.

15. L. Henkin. Completeness in the theory of types. The Journal of Symbolic Logic,
15(2):81–91, 1950.

16. W. H. Holliday, T. Hoshi, and T. F. Icard. Information dynamics and uniform
substitution. Synthese, 190:31–55, 2013.

17. E. Pacuit. Dynamic epistemic logic i: Modeling knowledge and belief. Philosophy
Compass, 8(9):798–814, 2013.

18. J. Plaza. Logics of public communications. Proceedings of the 4th International
Symposium on Methodologies for Intelligent Systems, 1989.

19. F. Raimondi and A. Lomuscio. Verification of multiagent systems via ordered
binary decision diagrams: an algorithm and its implementation. In Proceedings of
the Third International Joint Conference on Autonomous Agents and Multiagent
Systems, 2004. AAMAS 2004., pages 630–637. IEEE, 2004.

20. M. W. Tobias Nipkow, Lawrence C. Paulson. Isabelle/HOL: A Proof Assistant for
Higher-Order Logic. Springer-Verlag Berlin Heidelberg, 2002.

21. J. van Benthem, J. van Eijck, M. Gattinger, and K. Su. Symbolic model checking
for dynamic epistemic logic - S5 and beyond. J. Log. Comput., 28(2):367–402,
2018.

22. H. Van Ditmarsch, J. Y. Halpern, W. Van Der Hoek, and B. Kooi. An introduction
to logics of knowledge and belief. Handbook of epistemic logic, pages 1–51, 2015.

23. H. P. van Ditmarsch, W. Van Der Hoek, R. Van Der Meyden, and J. Ruan.
Model checking russian cards. Electronic Notes in Theoretical Computer Science,
149(2):105–123, 2006.

24. J. van Eijck. Demo—a demo of epistemic modelling. Interactive Logic.
Selected Papers from the 7th Augustus de Morgan Workshop, London,
http://homepages.cwi.nl/~jve/papers/07/pdfs/DEMO_IL.pdf, 2007.

25. J. van Eijck. Demo-s5. Tech. rep., CWI,
http://homepages.cwi.nl/~jve/software/demo_s5, 2014.

http://homepages.cwi.nl/~jve/papers/07/pdfs/DEMO_IL.pdf
http://homepages.cwi.nl/~jve/software/demo_s5

	Public Announcement Logic in HOL

